

F08KEF (SGBERD/DGBERD) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F08KEF (SGBERD/DGBERD) reduces a real m by n matrix to bidiagonal form.

2 Specification

```

SUBROUTINE F08KEF(M, N, A, LDA, D, E, TAUQ, TAUP, WORK, LWORK, INFO)
ENTRY      sgebrd(M, N, A, LDA, D, E, TAUQ, TAUP, WORK, LWORK, INFO)
INTEGER    M, N, LDA, LWORK, INFO
real      A(LDA,*), D(*), E(*), TAUQ(*), TAUP(*),
1          WORK(LWORK)

```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine reduces a real m by n matrix A to bidiagonal form B by an orthogonal transformation: $A = QB P^T$, where Q and P^T are orthogonal matrices of order m and n respectively.

If $m \geq n$, the reduction is given by:

$$A = Q \begin{pmatrix} B_1 \\ 0 \end{pmatrix} P^T = Q_1 B_1 P^T,$$

where B_1 is an n by n upper bidiagonal matrix and Q_1 consists of the first n columns of Q .

If $m < n$, the reduction is given by

$$A = Q(B_1 \ 0)P^T = QB_1 P_1^T,$$

where B_1 is an m by m lower bidiagonal matrix and P_1^T consists of the first m rows of P^T .

The orthogonal matrices Q and P are not formed explicitly but are represented as products of elementary reflectors (see the Chapter Introduction for details). Routines are provided to work with Q and P in this representation. (see Section 8).

4 References

- [1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

5 Parameters

- 1: M — INTEGER *Input*
On entry: m , the number of rows of the matrix A .
Constraint: $M \geq 0$.
- 2: N — INTEGER *Input*
On entry: n , the number of columns of the matrix A .
Constraint: $N \geq 0$.

- 3:** A(LDA,*) — *real* array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1,N)$.
On entry: the m by n matrix A .
On exit: if $m \geq n$, the diagonal and first super-diagonal are overwritten by the upper bidiagonal matrix B , elements below the diagonal are overwritten by details of the orthogonal matrix Q and elements above the first super-diagonal are overwritten by details of the orthogonal matrix P .
 If $m < n$, the diagonal and first sub-diagonal are overwritten by the lower bidiagonal matrix B , elements below the first sub-diagonal are overwritten by details of the orthogonal matrix Q and elements above the diagonal are overwritten by details of the orthogonal matrix P .
- 4:** LDA — INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08KEF (SGBERD/DGBERD) is called.
Constraint: $LDA \geq \max(1,M)$.
- 5:** D(*) — *real* array *Output*
Note: the dimension of the array D must be at least $\max(1,\min(M,N))$.
On exit: the diagonal elements of the bidiagonal matrix B .
- 6:** E(*) — *real* array *Output*
Note: the dimension of the array E must be at least $\max(1,\min(M,N)-1)$.
On exit: the off-diagonal elements of the bidiagonal matrix B .
- 7:** TAUQ(*) — *real* array *Output*
Note: the dimension of the array TAUQ must be at least $\max(1,\min(M,N))$.
On exit: further details of the orthogonal matrix Q .
- 8:** TAUP(*) — *real* array *Output*
Note: the dimension of the array TAUP must be at least $\max(1,\min(M,N))$.
On exit: further details of the orthogonal matrix P .
- 9:** WORK(LWORK) — *real* array *Workspace*
On exit: if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.
- 10:** LWORK — INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08KEF (SGBERD/DGBERD) is called.
Suggested value: for optimum performance LWORK should be at least $(M+N) \times nb$, where nb is the **blocksize**.
Constraint: $LWORK \geq \max(1,M,N)$.
- 11:** INFO — INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed bidiagonal form B satisfies $QBP^T = A + E$, where

$$\|E\|_2 \leq c(n)\epsilon \|A\|_2,$$

$c(n)$ is a modestly increasing function of n , and ϵ is the *machine precision*.

The elements of B themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the singular values and vectors.

8 Further Comments

The total number of floating-point operations is approximately $\frac{4}{3}n^2(3m - n)$ if $m \geq n$ or $\frac{4}{3}m^2(3n - m)$ if $m < n$.

If $m \gg n$, it can be more efficient to first call F08AEF (SGEQR/DGQR) to perform a QR factorization of A , and then to call this routine to reduce the factor R to bidiagonal form. This requires approximately $2n^2(m + n)$ floating-point operations.

If $m \ll n$, it can be more efficient to first call F08AHF (SGELQ/DGELQ) to perform an LQ factorization of A , and then to call this routine to reduce the factor L to bidiagonal form. This requires approximately $2m^2(m + n)$ operations.

To form the orthogonal matrices P^T and/or Q , this routine may be followed by calls to F08KFF (SORGBR/DORGBR):

to form the m by m orthogonal matrix Q

```
CALL SORGBR ('Q',M,M,N,A,LDA,TAUQ,WORK,LWORK,INFO)
```

but note that the second dimension of the array A must be at least M , which may be larger than was required by F08KEF;

to form the n by n orthogonal matrix P^T

```
CALL SORGBR ('P',N,N,M,A,LDA,TAUP,WORK,LWORK,INFO)
```

but note that the first dimension of the array A , specified by the parameter LDA , must be at least N , which may be larger than was required by F08KEF.

To apply Q or P to a real rectangular matrix C , this routine may be followed by a call to F08KGF (SORMBR/DORMBR).

The complex analogue of this routine is F08KSF (CGEBRD/ZGEBRD).

9 Example

To reduce the matrix A to bidiagonal form, where

$$A = \begin{pmatrix} -0.57 & -1.28 & -0.39 & 0.25 \\ -1.93 & 1.08 & -0.31 & -2.14 \\ 2.30 & 0.24 & 0.40 & -0.35 \\ -1.93 & 0.64 & -0.66 & 0.08 \\ 0.15 & 0.30 & 0.15 & -2.13 \\ -0.02 & 1.03 & -1.43 & 0.50 \end{pmatrix}.$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F08KEF Example Program Text
*      Mark 16 Release. MAG Copyright 1992.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, LDA, LWORK
      PARAMETER        (MMAX=8,NMAX=8,LDA=MMAX,LWORK=64*(MMAX+NMAX))
*      .. Local Scalars ..
      INTEGER          I, INFO, J, M, N
*      .. Local Arrays ..
      real            A(LDA,NMAX), D(NMAX), E(NMAX-1), TAUP(NMAX),
+                   TAUQ(NMAX), WORK(LWORK)
*      .. External Subroutines ..
      EXTERNAL         sgebrd
*      .. Intrinsic Functions ..
      INTRINSIC        MIN
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08KEF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.LE.MMAX .AND. N.LE.NMAX) THEN

*
*      Read A from data file
*
      READ (NIN,*) ((A(I,J),J=1,N),I=1,M)

*
*      Reduce A to bidiagonal form
*
      CALL sgebrd(M,N,A,LDA,D,E,TAUQ,TAUP,WORK,LWORK,INFO)

*
*      Print bidiagonal form
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Diagonal'
      WRITE (NOUT,99999) (D(I),I=1,MIN(M,N))
      IF (M.GE.N) THEN
         WRITE (NOUT,*) 'Super-diagonal'
      ELSE
         WRITE (NOUT,*) 'Sub-diagonal'
      END IF
      WRITE (NOUT,99999) (E(I),I=1,MIN(M,N)-1)
      END IF
      STOP
*
*
99999 FORMAT (1X,8F9.4)
      END

```

9.2 Program Data

```
F08KEF Example Program Data
  6  4                               :Values of M and N
-0.57 -1.28 -0.39  0.25
-1.93  1.08 -0.31 -2.14
  2.30  0.24  0.40 -0.35
-1.93  0.64 -0.66  0.08
  0.15  0.30  0.15 -2.13
-0.02  1.03 -1.43  0.50           :End of matrix A
```

9.3 Program Results

```
F08KEF Example Program Results
```

```
Diagonal
  3.6177  2.4161 -1.9213 -1.4265
Super-diagonal
  1.2587  1.5262 -1.1895
```
